

A Shorter Compiler Correctness Proof for Language IMP

Pasquale Noce
Software Engineer at HID Global, Italy
pasquale dot noce dot lavoro at gmail dot com
pasquale dot noce at hidglobal dot com

June 17, 2024

Abstract

This paper presents a compiler correctness proof for the didactic imperative programming language IMP, introduced in Nipkow and Klein’s book on formal programming language semantics (version of March 2021), whose size is just two thirds of the book’s proof in the number of formal text lines. As such, it promises to constitute a further enhanced reference for the formal verification of compilers meant for larger, real-world programming languages.

The presented proof does not depend on language determinism, so that the proposed approach can be applied to non-deterministic languages as well. As a confirmation, this paper extends IMP with an additional non-deterministic choice command, and proves compiler correctness, viz. the simulation of compiled code execution by source code, for such extended language.

Contents

1	Compiler formalization	1
1.1	Introduction	2
1.2	Definitions	2
2	Compiler correctness	6
2.1	Preliminary definitions and lemmas	6
2.2	Main theorem	11

1 Compiler formalization

```
theory Compiler
  imports
    HOL-IMP.BExp
```

HOL-IMP.Star
begin

1.1 Introduction

This paper presents a compiler correctness proof for the didactic imperative programming language IMP, introduced in [5], shorter than the proof described in [5] and included in the Isabelle2021 distribution [1]. Actually, the size of the presented proof is just two thirds of the book's proof in the number of formal text lines, and as such it promises to constitute a further enhanced reference for the formal verification of compilers meant for larger, real-world programming languages.

Given compiler *completeness*, viz. the simulation of source code execution by compiled code, "in a deterministic language like IMP", compiler correctness "reduces to preserving termination: if the machine program terminates, so must the source program", even though proving this "is not much easier" ([5], section 8.4). However, the presented proof does not depend on language determinism, so that the proposed approach is applicable to non-deterministic languages as well.

As a confirmation, this paper extends IMP with an additional command c_1 *OR* c_2 , standing for the non-deterministic choice between commands c_1 and c_2 , and proves compiler *correctness*, viz. the simulation of compiled code execution by source code, for such extended language. Of course, the afore-said comparison between proof sizes does not consider the lines in the proof of lemma *ccomp-correct* (which proves compiler correctness for commands) pertaining to non-deterministic choice, since this command is not included in the original language IMP. Anyway, non-deterministic choice turns out to extend that proof just by a modest number of lines.

For further information about the formal definitions and proofs contained in this paper, see Isabelle documentation, particularly [6], [4], [2], and [3].

1.2 Definitions

Here below are the definitions of IMP commands, extended with non-deterministic choice, as well as of their big-step semantics.

As in the original theory file [1], program counter's values are modeled using type *int* rather than *nat*. As a result, the same declarations and definitions used in [1] to deal with this modeling choice are adopted here as well.

```
declare [[coercion-enabled]]  
declare [[coercion int :: nat  $\Rightarrow$  int]]  
declare [[syntax-ambiguity-warning = false]]
```

```

datatype com =
  SKIP |
  Assign vname aexp (- ::= - [1000, 61] 61) |
  Seq com com (-;;/ - [60, 61] 60) |
  If bexp com com ((IF -/ THEN -/ ELSE -) [0, 0, 61] 61) |
  Or com com ((- OR -) [60, 61] 61) |
  While bexp com ((WHILE -/ DO -) [0, 61] 61)

inductive big-step :: com × state ⇒ state ⇒ bool (infix ⇒ 55) where
  Skip: (SKIP, s) ⇒ s |
  Assign: (x ::= a, s) ⇒ s(x := aval a s) |
  Seq: [(c1, s1) ⇒ s2; (c2, s2) ⇒ s3] ⇒ (c1;; c2, s1) ⇒ s3 |
  IfTrue: [bval b s; (c1, s) ⇒ t] ⇒ (IF b THEN c1 ELSE c2, s) ⇒ t |
  IfFalse: [¬ bval b s; (c2, s) ⇒ t] ⇒ (IF b THEN c1 ELSE c2, s) ⇒ t |
  Or1: (c1, s) ⇒ t ⇒ (c1 OR c2, s) ⇒ t |
  Or2: (c2, s) ⇒ t ⇒ (c1 OR c2, s) ⇒ t |
  WhileFalse: ¬ bval b s ⇒ (WHILE b DO c, s) ⇒ s |
  WhileTrue: [bval b s1; (c, s1) ⇒ s2; (WHILE b DO c, s2) ⇒ s3] ⇒
    (WHILE b DO c, s1) ⇒ s3

declare big-step.intros [intro]

abbreviation (output)
  isize xs ≡ int (length xs)

notation isize (size)

primrec (nonexhaustive) inth :: 'a list ⇒ int ⇒ 'a (infixl !! 100) where
  (x # xs) !! i = (if i = 0 then x else xs !! (i - 1))

lemma inth-append [simp]:
  0 ≤ i ⇒
    (xs @ ys) !! i = (if i < size xs then xs !! i else ys !! (i - size xs))
by (induction xs arbitrary: i, auto simp: algebra-simps)

```

Next, the instruction set and its semantics are defined. Particularly, to allow for the compilation of non-deterministic choice commands, the instruction set is extended with an additional instruction *JMPND* performing a non-deterministic jump – viz. as a result of its execution, the program counter unconditionally either jumps by the specified offset, or just moves to the next instruction.

As instruction execution can be non-deterministic, an inductively defined predicate *iexec*, rather than a simple non-recursive function as the one used in [1], must be introduced to define instruction semantics.

```

datatype instr =
  LOADI int | LOAD vname | ADD | STORE vname |

```

JMP int | *JMPLESS int* | *JMPGE int* | *JMPND int*

type-synonym *stack* = *val list*

type-synonym *config* = *int* × *state* × *stack*

abbreviation *hd2 xs* ≡ *hd (tl xs)*

abbreviation *tl2 xs* ≡ *tl (tl xs)*

inductive *iexec* :: *instr* × *config* ⇒ *config* ⇒ *bool* (**infix** ↦ 55) **where**

LoadI: (*LOADI i, pc, s, stk*) ↦ (*pc + 1, s, i # stk*) |

Load: (*LOAD x, pc, s, stk*) ↦ (*pc + 1, s, s x # stk*) |

Add: (*ADD, pc, s, stk*) ↦ (*pc + 1, s, (hd2 stk + hd stk) # tl2 stk*) |

Store: (*STORE x, pc, s, stk*) ↦ (*pc + 1, s(x := hd stk), tl stk*) |

Jmp: (*JMP i, pc, s, stk*) ↦ (*pc + i + 1, s, stk*) |

JmpLessY: *hd2 stk < hd stk* ⇒

(*JMPLESS i, pc, s, stk*) ↦ (*pc + i + 1, s, tl2 stk*) |

JmpLessN: *hd stk ≤ hd2 stk* ⇒

(*JMPLESS i, pc, s, stk*) ↦ (*pc + 1, s, tl2 stk*) |

JmpGeY: *hd stk ≤ hd2 stk* ⇒

(*JMPGE i, pc, s, stk*) ↦ (*pc + i + 1, s, tl2 stk*) |

JmpGeN: *hd2 stk < hd stk* ⇒

(*JMPGE i, pc, s, stk*) ↦ (*pc + 1, s, tl2 stk*) |

JmpNdY: (*JMPND i, pc, s, stk*) ↦ (*pc + i + 1, s, stk*) |

JmpNdN: (*JMPND i, pc, s, stk*) ↦ (*pc + 1, s, stk*)

declare *iexec.intros* [*intro*]

inductive-cases *LoadIE* [*elim!*]: (*LOADI i, pc, s, stk*) ↦ *cf*

inductive-cases *LoadE* [*elim!*]: (*LOAD x, pc, s, stk*) ↦ *cf*

inductive-cases *AddE* [*elim!*]: (*ADD, pc, s, stk*) ↦ *cf*

inductive-cases *StoreE* [*elim!*]: (*STORE x, pc, s, stk*) ↦ *cf*

inductive-cases *JmpE* [*elim!*]: (*JMP i, pc, s, stk*) ↦ *cf*

inductive-cases *JmpLessE* [*elim!*]: (*JMPLESS i, pc, s, stk*) ↦ *cf*

inductive-cases *JmpGeE* [*elim!*]: (*JMPGE i, pc, s, stk*) ↦ *cf*

inductive-cases *JmpNdE* [*elim!*]: (*JMPND i, pc, s, stk*) ↦ *cf*

definition *exec1* :: *instr list* ⇒ *config* ⇒ *config* ⇒ *bool*

((-/ ⊢ / -/ → / -) 55) **where**

P ⊢ *cf* → *cf'* ≡ (*P* !! *fst cf, cf*) ↦ *cf' ∧ 0 ≤ fst cf ∧ fst cf < size P*

abbreviation *exec* :: *instr list* ⇒ *config* ⇒ *config* ⇒ *bool*

((-/ ⊢ / -/ → * / -) 55) **where**

exec P ≡ *star (exec1 P)*

Next, compilation is formalized for arithmetic and boolean expressions (functions *acomp* and *bcomp*), as well as for commands (function *ccomp*). Particularly, as opposed to what happens in [1], here *bcomp* takes a single input, viz. a 3-tuple comprised of a boolean expression, a flag, and a jump off-

set. In this way, all three functions accept a single input, which enables to streamline the compiler correctness proof developed in what follows.

```

primrec acomp :: aexp  $\Rightarrow$  instr list where
acomp (N i) = [LOADI i] |
acomp (V x) = [LOAD x] |
acomp (Plus a1 a2) = acomp a1 @ acomp a2 @ [ADD]

fun bcomp :: bexp  $\times$  bool  $\times$  int  $\Rightarrow$  instr list where
bcomp (Bc v, f, i) = (if v = f then [JMP i] else []) |
bcomp (Not b, f, i) = bcomp (b,  $\neg$  f, i) |
bcomp (And b1 b2, f, i) =
  (let cb2 = bcomp (b2, f, i);
    cb1 = bcomp (b1, False, size cb2 + (if f then 0 else i))
  in cb1 @ cb2) |
bcomp (Less a1 a2, f, i) =
  acomp a1 @ acomp a2 @ (if f then [JMPLESS i] else [JMPGE i])

```

```

primrec ccomp :: com  $\Rightarrow$  instr list where
ccomp SKIP = [] |
ccomp (x ::= a) = acomp a @ [STORE x] |
ccomp (c1;; c2) = ccomp c1 @ ccomp c2 |
ccomp (IF b THEN c1 ELSE c2) =
  (let cc1 = ccomp c1; cc2 = ccomp c2; cb = bcomp (b, False, size cc1 + 1)
  in cb @ cc1 @ JMP (size cc2) # cc2) |
ccomp (c1 OR c2) =
  (let cc1 = ccomp c1; cc2 = ccomp c2
  in JMPND (size cc1 + 1) # cc1 @ JMP (size cc2) # cc2) |
ccomp (WHILE b DO c) =
  (let cc = ccomp c; cb = bcomp (b, False, size cc + 1)
  in cb @ cc @ [JMP (- (size cb + size cc + 1))])

```

Finally, two lemmas are proven automatically (both seem not to be included in the standard library, though being quite basic) and registered for use by automatic proof tactics. In more detail:

- The former lemma is an elimination rule similar to *impCE*, with the difference that it retains the antecedent of the implication in the premise where the consequent is assumed to hold. This rule permits to have both assumptions \neg *bval* *b* *s* and *bval* *b* *s* in the respective cases resulting from the execution of boolean expression *b* in state *s*.
- The latter one is an introduction rule similar to *Suc-lessI*, with the difference that its second assumption is more convenient for proving statements of the form *Suc* *m* < *n* arising from the compiler correctness proof developed in what follows.

lemma *impCE2* [*elim!*]:
 $\llbracket P \longrightarrow Q; \neg P \Longrightarrow R; P \Longrightarrow Q \Longrightarrow R \rrbracket \Longrightarrow R$
by *blast*

lemma *Suc-lessI2* [*intro!*]:
 $\llbracket m < n; m \neq n - 1 \rrbracket \Longrightarrow \text{Suc } m < n$
by *simp*

end

2 Compiler correctness

theory *Compiler2*
imports *Compiler*
begin

2.1 Preliminary definitions and lemmas

Now everything is ready for the compiler correctness proof. First, two predicates are introduced, *execl* and *execl-all*, both taking as inputs a program, i.e. a list of instructions, P and a list of program configurations cfs , and respectively denoted using notations $P \models cfs$ and $P \models cfs\Box$. In more detail:

- $P \models cfs$ means that program P *may* transform each configuration within cfs into the subsequent one, if any (word *may* reflects the fact that programs can be non-deterministic in this case study).
Thus, *execl* formalizes the notion of a *small-step* program execution.
- $P \models cfs\Box$ reinforces $P \models cfs$ by additionally requiring that cfs be nonempty, the initial program counter be zero (viz. execution starts from the first instruction in P), and the final program counter falls outside P (viz. execution terminates).
Thus, *execl-all* formalizes the notion of a *complete small-step* program execution, so that assumptions $acomp\ a \models cfs\Box$, $bcomp\ x \models cfs\Box$, $ccomp\ c \models cfs\Box$ will be used in the compiler correctness proofs for arithmetic/boolean expressions and commands.

Moreover, predicates *apred*, *bpred*, and *cpred* are defined to capture the link between the initial and the final configuration upon the execution of an arithmetic expression, a boolean expression, and a whole program, respectively, and abbreviation *off* is introduced as a commodity to shorten the subsequent formal text.

fun *execl* :: *instr list* \Rightarrow *config list* \Rightarrow *bool* (**infix** \models 55) **where**

$$P \models cf \# cf' \# cfs = (P \vdash cf \rightarrow cf' \wedge P \models cf' \# cfs) \mid$$

$$P \models - = True$$

definition *execl-all* :: *instr list* \Rightarrow *config list* \Rightarrow *bool* ((-/ \models / - \square) 55) **where**
 $P \models cfs \square \equiv P \models cfs \wedge cfs \neq [] \wedge$
 $fst (cfs ! 0) = 0 \wedge fst (cfs ! (length cfs - 1)) \notin \{0..<size P\}$

definition *apred* :: *aexp* \Rightarrow *config* \Rightarrow *config* \Rightarrow *bool* **where**
 $apred \equiv \lambda a (pc, s, stk) (pc', s', stk')$.
 $pc' = pc + size (acomp a) \wedge s' = s \wedge stk' = aval a s \# stk$

definition *bpred* :: *bexp* \times *bool* \times *int* \Rightarrow *config* \Rightarrow *config* \Rightarrow *bool* **where**
 $bpred \equiv \lambda (b, f, i) (pc, s, stk) (pc', s', stk')$.
 $pc' = pc + size (bcomp (b, f, i)) + (if bval b s = f then i else 0) \wedge$
 $s' = s \wedge stk' = stk$

definition *cpred* :: *com* \Rightarrow *config* \Rightarrow *config* \Rightarrow *bool* **where**
 $cpred \equiv \lambda c (pc, s, stk) (pc', s', stk')$.
 $pc' = pc + size (ccomp c) \wedge (c, s) \Rightarrow s' \wedge stk' = stk$

abbreviation *off* :: *instr list* \Rightarrow *config* \Rightarrow *config* **where**
 $off P cf \equiv (fst cf - size P, snd cf)$

Next, some lemmas about *execl* and *execl-all* are proven. In more detail, given a program P and a list of configurations cfs such that $P \models cfs$:

- Lemma *execl-next* states that for any configuration in cfs but the last one, the subsequent configuration must result from the execution of the referenced instruction of P in that configuration. Thus, *execl-next* permits to reproduce the execution of a single instruction.
- Lemma *execl-last* states that a configuration in cfs whose program counter falls outside P must be the last one in cfs . Thus, *execl-last* permits to infer the completion of program execution.
- Lemma *execl-drop* states that $P \models drop\ n\ cfs$ for any natural number n , and will be used to prove compiler correctness for loops by induction over the length of the list of configurations cfs .

Furthermore, some other lemmas enabling to prove compiler correctness automatically for constructors N , V (arithmetic expressions), Bc (boolean expressions) and $SKIP$ (commands) are also proven.

lemma *iexec-offset-aux*:
 $(i :: int) + 1 - j = i - j + 1 \wedge i + j - k + 1 = i - k + j + 1$

by *simp*

lemma *iexec-offset* [intro]:

$(ins, pc, s, stk) \mapsto (pc', s', stk') \implies$
 $(ins, pc - i, s, stk) \mapsto (pc' - i, s', stk')$

by (*erule iexec.cases, auto simp: iexec-offset-aux*)

lemma *execl-next*:

$\llbracket P \models cfs; k < \text{length } cfs; k \neq \text{length } cfs - 1 \rrbracket \implies$
 $(P \text{ !! } fst (cfs ! k), cfs ! k) \mapsto cfs ! Suc k$

by (*induction cfs arbitrary: k rule: execl.induct, auto simp: nth-Cons exec1-def split: nat.split*)

lemma *execl-last*:

$\llbracket P \models cfs; k < \text{length } cfs; fst (cfs ! k) \notin \{0..<\text{size } P\} \rrbracket \implies$
 $\text{length } cfs - 1 = k$

by (*induction cfs arbitrary: k rule: execl.induct, auto simp: nth-Cons exec1-def split: nat.split-asm*)

lemma *execl-drop*:

$P \models cfs \implies P \models \text{drop } n \text{ } cfs$

by (*induction cfs arbitrary: n rule: execl.induct, simp-all add: drop-Cons split: nat.split*)

lemma *execl-all-N* [simplified, intro]:

$\llbracket LOADI i \rrbracket \models cfs \square \implies \text{apred } (N i) (cfs ! 0) (cfs ! (\text{length } cfs - 1))$

by (*clarsimp simp: execl-all-def apred-def, cases cfs ! 0, subgoal-tac length cfs - 1 = 1, frule-tac [] execl-next, ((rule ccontr)?, fastforce, (rule execl-last)?)+*)

lemma *execl-all-V* [simplified, intro]:

$\llbracket LOAD x \rrbracket \models cfs \square \implies \text{apred } (V x) (cfs ! 0) (cfs ! (\text{length } cfs - 1))$

by (*clarsimp simp: execl-all-def apred-def, cases cfs ! 0, subgoal-tac length cfs - 1 = 1, frule-tac [] execl-next, ((rule ccontr)?, fastforce, (rule execl-last)?)+*)

lemma *execl-all-Bc* [simplified, intro]:

$\llbracket \text{if } v = f \text{ then } [JMP i] \text{ else } [] \rrbracket \models cfs \square; 0 \leq i \implies$
 $\text{bpred } (Bc v, f, i) (cfs ! 0) (cfs ! (\text{length } cfs - 1))$

by (*clarsimp simp: execl-all-def bpred-def split: if-split-asm, cases cfs ! 0, subgoal-tac length cfs - 1 = 1, frule-tac [1-2] execl-next, ((rule ccontr)?, force, (rule execl-last)?)+, rule execl.cases [of ([], cfs)], auto simp: execl-def*)

lemma *execl-all-SKIP* [simplified, intro]:

$[] \models cfs \square \implies \text{cpred } SKIP (cfs ! 0) (cfs ! (\text{length } cfs - 1))$

by (*rule execl.cases [of ([], cfs)], auto simp: execl-all-def execl-def cpred-def*)

Next, lemma *execl-all-sub* is proven. It states that, if $P @ P' x @ P'' \models cfs \square$, configuration cf within cfs refers to the start of program $P' x$, and the initial and the final configuration in every complete execution of $P' x$ satisfy predicate $Q x$, then there exists a configuration cf' in cfs such that cf and cf' satisfy $Q x$.

Thus, this lemma permits to reproduce the execution of a subprogram, particularly:

- a compiled arithmetic expression a , where $Q = apred$ and $x = a$,
- a compiled boolean expression b , where $Q = bpred$ and $x = (b, f, i)$ (given a flag f and a jump offset i), and
- a compiled command c , where $Q = cpred$ and $x = c$.

Furthermore, lemma *execl-all-sub2* is derived from *execl-all-sub* to enable a shorter symbolical execution of two consecutive subprograms.

lemma *execl-sub-aux*:

$\llbracket \bigwedge m n. \forall k \in \{m..<n\}. Q P' (((pc, s, stk) \# cfs) ! k) \implies P' \models$
 $\text{map } (off P) (\text{case } m \text{ of } 0 \Rightarrow (pc, s, stk) \# \text{take } n \text{ cfs} \mid \text{Suc } m \Rightarrow F \text{ cfs } m \ n);$
 $\forall k \in \{m..<n+m+\text{length } cfs'\}. Q P' ((cfs' @ (pc, s, stk) \# cfs) ! (k-m)) \rrbracket \implies$
 $P' \models (pc - \text{size } P, s, stk) \# \text{map } (off P) (\text{take } n \text{ cfs})$
 $(\text{is } [\bigwedge - \cdot. \forall k \in -. Q P' (?F k) \implies -; \forall k \in ?A. Q P' (?G k)] \implies -)$
by (*subgoal-tac* $\forall k \in \{0..<n\}. Q P' (?F k)$, *fastforce*, *subgoal-tac*
 $\forall k \in \{0..<n\}. k + m + \text{length } cfs' \in ?A \wedge ?F k = ?G (k + m + \text{length } cfs')$,
fastforce, *simp add: nth-append*)

lemma *execl-sub*:

$\llbracket P @ P' @ P'' \models cfs; \forall k \in \{m..<n\}. \text{size } P \leq \text{fst } (cfs ! k) \wedge \text{fst } (cfs ! k) - \text{size } P < \text{size } P' \rrbracket \implies$
 $P' \models \text{map } (off P) (\text{drop } m (\text{take } (\text{Suc } n) \text{ cfs}))$
 $(\text{is } [-; \forall k \in -. ?P P' cfs k] \implies P' \models \text{map } - (?F \text{ cfs } m (\text{Suc } n)))$

proof (*induction cfs arbitrary: m n rule: execl.induct [of - P']*, *auto*
simp: take-Cons drop-Cons exec1-def split: nat.split, force, force, force,
erule execl-sub-aux [where m = 0], subst append-Cons [of - []], simp,
erule execl-sub-aux [where m = Suc 0 and cfs' = []], simp)

fix $P' pc s stk cfs m n$

let $?cf = (pc, s, stk) :: \text{config}$

assume $\bigwedge m n. \forall k \in \{m..<n\}. ?P P' (?cf \# cfs) k \implies P' \models$

$\text{map } (off P) (\text{case } m \text{ of } 0 \Rightarrow ?cf \# \text{take } n \text{ cfs} \mid \text{Suc } m \Rightarrow ?F \text{ cfs } m \ n)$

moreover assume $\forall k \in \{\text{Suc } (\text{Suc } m)..<\text{Suc } n\}. ?P P' cfs (k - \text{Suc } (\text{Suc } 0))$

hence $\forall k \in \{\text{Suc } m..<n\}. ?P P' (?cf \# cfs) k$

by *force*

ultimately show $P' \models \text{map } (off P) (?F \text{ cfs } m \ n)$

by *fastforce*

qed

lemma *execl-all-sub* [rule-format]:

assumes

$A: P @ P' x @ P'' \models \text{cfs}\square$ **and**

$B: k < \text{length } \text{cfs}$ **and**

$C: \text{fst } (\text{cfs} ! k) = \text{size } P$ **and**

$D: \forall \text{cfs}. P' x \models \text{cfs}\square \longrightarrow Q x (\text{cfs} ! 0) (\text{cfs} ! (\text{length } \text{cfs} - 1))$

shows $\exists k' < \text{length } \text{cfs}. Q x (\text{off } P (\text{cfs} ! k)) (\text{off } P (\text{cfs} ! k'))$

proof –

let $?P = \lambda k. \text{size } P \leq \text{fst } (\text{cfs} ! k) \wedge \text{fst } (\text{cfs} ! k) - \text{size } P < \text{size } (P' x)$

let $?A = \{k'. k' \in \{k..<\text{length } \text{cfs}\} \wedge \neg ?P k'\}$

have $E: \text{Min } ?A \in ?A$

using A **and** B **by** (*rule-tac Min-in, simp-all add: execl-all-def,*

rule-tac exI [of - length cfs - 1], auto)

hence $\text{map } (\text{off } P) (\text{drop } k (\text{take } (\text{Suc } (\text{Min } ?A)) \text{cfs})) ! 0 = \text{off } P (\text{cfs} ! k)$

(**is** $?cfs ! - = -$)

by *auto*

moreover **have** $\text{length } \text{cfs} \leq \text{Suc } (\text{Min } ?A) \longrightarrow \text{Min } ?A = \text{length } \text{cfs} - 1$

using E **by** *auto*

with A **and** B **and** E **have** $F: ?cfs ! (\text{length } ?cfs - 1) = \text{off } P (\text{cfs} ! \text{Min } ?A)$

by (*subst nth-map, auto simp: min-def execl-all-def, arith*)

hence $?cfs \neq [] \wedge \text{fst } (?cfs ! (\text{length } ?cfs - 1)) \notin \{0..<\text{size } (P' x)\}$

using E **by** (*auto simp: min-def*)

moreover **have** $\neg (\exists k'. k' \in \{k'. k' \in \{k..<\text{Min } ?A\} \wedge \neg ?P k'\})$

by (*rule notI, erule exE, frule rev-subsetD [of - - ?A], rule subsetI,*

insert E, simp, subgoal-tac finite ?A, drule Min-le, force+)

hence $P' x \models ?cfs$

using A **by** (*subst (asm) execl-all-def, rule-tac execl-sub, blast+*)

ultimately **have** $Q x (?cfs ! 0) (?cfs ! (\text{length } ?cfs - 1))$

using C **and** D **by** (*auto simp: execl-all-def*)

thus *?thesis*

using E **and** F **by** (*rule-tac exI [of - Min ?A], auto*)

qed

lemma *execl-all-sub2*:

assumes

$A: P x @ P' x' @ P'' \models \text{cfs}\square$

(**is** $?P \models -\square$) **and**

$B: \bigwedge \text{cfs}. P x \models \text{cfs}\square \implies (\lambda(\text{pc}, s, \text{stk}) (\text{pc}', s', \text{stk}').$

$\text{pc}' = \text{pc} + \text{size } (P x) + I s \wedge Q s s' \wedge \text{stk}' = F s \text{stk}$

$(\text{cfs} ! 0) (\text{cfs} ! (\text{length } \text{cfs} - 1))$

(**is** $\bigwedge \text{cfs}. - \implies ?Q x (\text{cfs} ! 0) (\text{cfs} ! (\text{length } \text{cfs} - 1))$) **and**

$C: \bigwedge \text{cfs}. P' x' \models \text{cfs}\square \implies (\lambda(\text{pc}, s, \text{stk}) (\text{pc}', s', \text{stk}').$

$\text{pc}' = \text{pc} + \text{size } (P' x') + I' s \wedge Q' s s' \wedge \text{stk}' = F' s \text{stk}$

$(\text{cfs} ! 0) (\text{cfs} ! (\text{length } \text{cfs} - 1))$

(**is** $\bigwedge \text{cfs}. - \implies ?Q' x' (\text{cfs} ! 0) (\text{cfs} ! (\text{length } \text{cfs} - 1))$) **and**

$D: I (\text{fst } (\text{snd } (\text{cfs} ! 0))) = 0$

shows $\exists k < \text{length } \text{cfs}. \exists t. (\lambda(\text{pc}, s, \text{stk}) (\text{pc}', s', \text{stk}').$

$\text{pc} = 0 \wedge \text{pc}' = \text{size } (P x) + \text{size } (P' x') + I' t \wedge Q s t \wedge Q' t s' \wedge$

$\text{stk}' = F' t (F s \text{stk})) (\text{cfs} ! 0) (\text{cfs} ! k)$

by (subgoal-tac [] @ ?P \models cfs \square , drule execl-all-sub [where k = 0 and Q = ?Q], insert A B, (clarsimp simp: execl-all-def)+, insert A C D, drule execl-all-sub [where Q = ?Q], simp+, clarify, rule exI, rule conjI, simp, rule exI, auto)

2.2 Main theorem

It is time to prove compiler correctness. First, lemmas *acom-p-acom*, *bcomp-bcomp* are derived from *execl-all-sub2* to reproduce the execution of two consecutive compiled arithmetic expressions (possibly generated by both *acom-p* and *bcomp*) and boolean expressions (possibly generated by *bcomp*), respectively. Subsequently, the correctness of *acom-p* and *bcomp* is proven in lemmas *acom-p-correct*, *bcomp-correct*.

lemma *acom-p-acom*:

[[*acom-p* a_1 @ *acom-p* a_2 @ P \models cfs \square ;
 \wedge cfs. *acom-p* a_1 \models cfs \square \implies *apred* a_1 (cfs ! 0) (cfs ! (length cfs - 1));
 \wedge cfs. *acom-p* a_2 \models cfs \square \implies *apred* a_2 (cfs ! 0) (cfs ! (length cfs - 1))] \implies
 case cfs ! 0 of (pc, s, stk) \Rightarrow pc = 0 \wedge ($\exists k < \text{length cfs}$. cfs ! k =
 (size (*acom-p* a_1 @ *acom-p* a_2), s, *aval* a_2 s # *aval* a_1 s # stk))
 by (drule execl-all-sub2 [where I = λs . 0 and I' = λs . 0 and Q = λs s'. s' = s
 and Q' = λs s'. s' = s and F = λs stk. *aval* a_1 s # stk
 and F' = λs stk. *aval* a_2 s # stk], auto simp: *apred-def*)

lemma *bcomp-bcomp*:

[[*bcomp* (b_1, f_1, i_1) @ *bcomp* (b_2, f_2, i_2) \models cfs \square ;
 \wedge cfs. *bcomp* (b_1, f_1, i_1) \models cfs \square \implies
bpred (b_1, f_1, i_1) (cfs ! 0) (cfs ! (length cfs - 1));
 \wedge cfs. *bcomp* (b_2, f_2, i_2) \models cfs \square \implies
bpred (b_2, f_2, i_2) (cfs ! 0) (cfs ! (length cfs - 1))] \implies
 case cfs ! 0 of (pc, s, stk) \Rightarrow pc = 0 \wedge (bval b_1 s \neq $f_1 \longrightarrow$
 ($\exists k < \text{length cfs}$. cfs ! k = (size (*bcomp* (b_1, f_1, i_1) @ *bcomp* (b_2, f_2, i_2)) +
 (if bval b_2 s = f_2 then i_2 else 0), s, stk)))
 by (clarify, rule conjI, simp add: execl-all-def, rule impI, subst (asm) append-Nil2
 [symmetric], drule execl-all-sub2 [where I = λs . if bval b_1 s = f_1 then i_1 else 0
 and I' = λs . if bval b_2 s = f_2 then i_2 else 0 and Q = λs s'. s' = s
 and Q' = λs s'. s' = s and F = λs stk. stk and F' = λs stk. stk],
 auto simp: *bpred-def*)

lemma *acom-p-correct* [simplified, intro]:

acom-p a \models cfs \square \implies *apred* a (cfs ! 0) (cfs ! (length cfs - 1))
proof (induction a arbitrary: cfs, simp-all, frule-tac [3] *acom-p-acom*, auto)
 fix a_1 a_2 cfs s stk k
 assume A: *acom-p* a_1 @ *acom-p* a_2 @ [ADD] \models cfs \square
 (is ?ca₁ @ ?ca₂ @ ?i \models - \square)
 assume B: k < length cfs and
 C: cfs ! k = (size ?ca₁ + size ?ca₂, s, *aval* a_2 s # *aval* a_1 s # stk)
 hence cfs ! Suc k = (size (?ca₁ @ ?ca₂ @ ?i), s, *aval* (Plus a_1 a_2) s # stk)
 using A by (insert execl-next [of ?ca₁ @ ?ca₂ @ ?i cfs k],

simp add: execl-all-def, drule-tac impI, auto)
thus *apred (Plus a₁ a₂) (0, s, stk) (cfs ! (length cfs - Suc 0))*
using *A and B and C by (insert execl-last [of ?ca₁ @ ?ca₂ @ ?i cfs Suc k],*
simp add: execl-all-def apred-def, drule-tac impI, auto)
qed

lemma *bcomp-correct [simplified, intro]:*
 $\llbracket bcomp\ x \models cfs\Box; 0 \leq snd\ (snd\ x) \rrbracket \implies bpred\ x\ (cfs\ !\ 0)\ (cfs\ !\ (length\ cfs\ -\ 1))$
proof (*induction x arbitrary: cfs rule: bcomp.induct, simp-all add: Let-def,*
frule-tac [4] acomp-acomp, frule-tac [3] bcomp-bcomp, auto, force simp: bpred-def)
fix *b₁ b₂ f i cfs s stk*
assume *A: bcomp (b₁, False, size (bcomp (b₂, f, i)) + (if f then 0 else i)) @*
bcomp (b₂, f, i) \models cfs \Box
(is bcomp (-, -, ?n + ?i) @ ?cb \models - \Box)
moreover assume *B: cfs ! 0 = (0, s, stk) and*
 $\bigwedge cb\ cfs.\ \llbracket cb = ?cb; bcomp\ (b_1,\ False,\ ?n + ?i) \models cfs\Box \rrbracket \implies$
 $bpred\ (b_1,\ False,\ ?n + ?i)\ (cfs\ !\ 0)\ (cfs\ !\ (length\ cfs\ -\ Suc\ 0))$
ultimately have $\exists k < length\ cfs.\ bpred\ (b_1,\ False,\ ?n + ?i)$
(off [] (cfs ! 0)) (off [] (cfs ! k))
by (*rule-tac execl-all-sub, auto simp: execl-all-def*)
moreover assume *C: \neg bval b₁ s*
ultimately obtain *k where D: k < length cfs and*
E: cfs ! k = (size (bcomp (b₁, False, ?n + ?i)) + ?n + ?i, s, stk)
using *B by (auto simp: bpred-def)*
assume $0 \leq i$
thus *bpred (And b₁ b₂, f, i) (0, s, stk) (cfs ! (length cfs - Suc 0))*
using *A and C and D and E by (insert execl-last, auto simp:*
execl-all-def bpred-def Let-def)
next
fix *b₁ b₂ f i cfs s stk k*
assume *A: bcomp (b₁, False, size (bcomp (b₂, f, i)) + (if f then 0 else i)) @*
bcomp (b₂, f, i) \models cfs \Box
(is ?cb₁ @ ?cb₂ \models - \Box)
assume *k < length cfs and 0 \leq i and bval b₁ s and*
 $cfs\ !\ k = (size\ ?cb_1 + size\ ?cb_2 + (if\ bval\ b_2\ s = f\ then\ i\ else\ 0),\ s,\ stk)$
thus *bpred (And b₁ b₂, f, i) (0, s, stk) (cfs ! (length cfs - Suc 0))*
using *A by (insert execl-last, auto simp: execl-all-def bpred-def Let-def)*
next
fix *a₁ a₂ f i cfs s stk k*
assume *A: acomp a₁ @ acomp a₂ @*
(if f then [JMPLESS i] else [JMPGE i]) \models cfs \Box
(is ?ca₁ @ ?ca₂ @ ?i \models - \Box)
assume *B: k < length cfs and*
 $C: cfs\ !\ k = (size\ ?ca_1 + size\ ?ca_2,\ s,\ aval\ a_2\ s \#\ aval\ a_1\ s \#\ stk)$
hence *D: cfs ! Suc k =*
 $(size\ (?ca_1\ @\ ?ca_2\ @\ ?i) + (if\ bval\ (Less\ a_1\ a_2)\ s = f\ then\ i\ else\ 0),\ s,\ stk)$
using *A by (insert execl-next [of ?ca₁ @ ?ca₂ @ ?i cfs k],*
simp add: execl-all-def, drule-tac impI, auto split: if-split-asm)
assume $0 \leq i$

with A **and** B **and** C **and** D **have** $\text{length } cfs - 1 = \text{Suc } k$
by (*rule-tac execl-last, auto simp: execl-all-def, simp split: if-split-asm*)
thus $\text{bpred } (\text{Less } a_1 a_2, f, i) (0, s, stk) (cfs ! (\text{length } cfs - \text{Suc } 0))$
using D **by** (*simp add: bpred-def*)
qed

Next, lemmas *bcomp-ccomp*, *ccomp-ccomp* are derived to reproduce the execution of a compiled boolean expression followed by a compiled command and of two consecutive compiled commands, respectively (possibly generated by *ccomp*). Then, compiler correctness for loops and for all commands is proven in lemmas *while-correct* and *ccomp-correct*, respectively by induction over the length of the list of configurations and by structural induction over commands.

lemma *bcomp-ccomp*:

$\llbracket \text{bcomp } (b, f, i) @ \text{ccomp } c @ P \models \text{cfs}\square; 0 \leq i; \wedge \text{cfs. } \text{ccomp } c \models \text{cfs}\square \implies \text{cpred } c (cfs ! 0) (cfs ! (\text{length } cfs - 1)) \rrbracket \implies$
 $\text{case } cfs ! 0 \text{ of } (pc, s, stk) \Rightarrow pc = 0 \wedge (\text{bval } b \ s \neq f \longrightarrow$
 $(\exists k < \text{length } cfs. \text{case } cfs ! k \text{ of } (pc', s', stk') \Rightarrow$
 $pc' = \text{size } (\text{bcomp } (b, f, i) @ \text{ccomp } c) \wedge (c, s) \Rightarrow s' \wedge \text{stk}' = \text{stk}))$
by (*clarify, rule conjI, simp add: execl-all-def, rule impI, drule execl-all-sub2*
 $[\text{where } I = \lambda s. \text{if } \text{bval } b \ s = f \text{ then } i \text{ else } 0 \text{ and } I' = \lambda s. 0$
 $\text{and } Q = \lambda s \ s'. s' = s \text{ and } Q' = \lambda s \ s'. (c, s) \Rightarrow s' \text{ and } F = \lambda s \ \text{stk. } \text{stk}$
 $\text{and } F' = \lambda s \ \text{stk. } \text{stk}], \text{insert } \text{bcomp-correct } [\text{of } (b, f, i)],$
auto simp: bpred-def cpred-def)

lemma *ccomp-ccomp*:

$\llbracket \text{ccomp } c_1 @ \text{ccomp } c_2 \models \text{cfs}\square; \wedge \text{cfs. } \text{ccomp } c_1 \models \text{cfs}\square \implies \text{cpred } c_1 (cfs ! 0) (cfs ! (\text{length } cfs - 1));$
 $\wedge \text{cfs. } \text{ccomp } c_2 \models \text{cfs}\square \implies \text{cpred } c_2 (cfs ! 0) (cfs ! (\text{length } cfs - 1)) \rrbracket \implies$
 $\text{case } cfs ! 0 \text{ of } (pc, s, stk) \Rightarrow pc = 0 \wedge (\exists k < \text{length } cfs. \exists t.$
 $\text{case } cfs ! k \text{ of } (pc', s', stk') \Rightarrow pc' = \text{size } (\text{ccomp } c_1 @ \text{ccomp } c_2) \wedge$
 $(c_1, s) \Rightarrow t \wedge (c_2, t) \Rightarrow s' \wedge \text{stk}' = \text{stk})$
by (*subst (asm) append-Nil2 [symmetric], drule execl-all-sub2 [where* $I = \lambda s. 0$
 $\text{and } I' = \lambda s. 0 \text{ and } Q = \lambda s \ s'. (c_1, s) \Rightarrow s' \text{ and } Q' = \lambda s \ s'. (c_2, s) \Rightarrow s'$
 $\text{and } F = \lambda s \ \text{stk. } \text{stk} \text{ and } F' = \lambda s \ \text{stk. } \text{stk}], \text{auto simp: cpred-def, force}$)

lemma *while-correct* [*simplified, intro*]:

$\llbracket \text{bcomp } (b, \text{False}, \text{size } (\text{ccomp } c) + 1) @ \text{ccomp } c @$
 $[\text{JMP } (- (\text{size } (\text{bcomp } (b, \text{False}, \text{size } (\text{ccomp } c) + 1) @ \text{ccomp } c) + 1))] \models$
 $\text{cfs}\square;$
 $\wedge \text{cfs. } \text{ccomp } c \models \text{cfs}\square \implies \text{cpred } c (cfs ! 0) (cfs ! (\text{length } cfs - 1)) \rrbracket \implies$
 $\text{cpred } (\text{WHILE } b \ \text{DO } c) (cfs ! 0) (cfs ! (\text{length } cfs - \text{Suc } 0))$
 $(\text{is } \llbracket ?cb @ ?cc @ [\text{JMP } (- ?n)] \models \neg\square; \wedge \neg. \neg \implies \neg \rrbracket \implies ?Q \ \text{cfs})$
proof (*induction cfs rule: length-induct, frule bcomp-ccomp, auto*)
fix $cfs \ s \ \text{stk}$
assume $A: ?cb @ ?cc @ [\text{JMP } (- \text{size } ?cb - \text{size } ?cc - 1)] \models \text{cfs}\square$

hence $\exists k < \text{length } cfs. \text{bpred } (b, \text{False}, \text{size } (c\text{comp } c) + 1)$
 $(\text{off } [] (cfs ! 0)) (\text{off } [] (cfs ! k))$
by (*rule-tac execl-all-sub, auto simp: execl-all-def*)
moreover assume $B: \neg \text{bval } b \text{ s and } cfs ! 0 = (0, s, stk)$
ultimately obtain k **where** $k < \text{length } cfs$ **and** $cfs ! k = (?n, s, stk)$
by (*auto simp: bpred-def*)
thus $c\text{pred } (\text{WHILE } b \text{ DO } c) (0, s, stk) (cfs ! (\text{length } cfs - \text{Suc } 0))$
using A **and** B **by** (*insert execl-last, auto simp: execl-all-def cpred-def Let-def*)
next
fix $cfs \text{ s } s' \text{ stk } k$
assume $A: ?cb @ ?cc @ [\text{JMP } (- \text{size } ?cb - \text{size } ?cc - 1)] \models cfs \square$
 $(\text{is } ?P \models -\square)$
assume $B: k < \text{length } cfs$ **and** $cfs ! k = (\text{size } ?cb + \text{size } ?cc, s', stk)$
moreover from this have $C: k \neq \text{length } cfs - 1$
using A **by** (*rule-tac notI, simp add: execl-all-def*)
ultimately have $D: cfs ! \text{Suc } k = (0, s', stk)$
using A **by** (*insert execl-next [of ?P cfs k], auto simp: execl-all-def*)
moreover have $E: \text{Suc } k + (\text{length } (\text{drop } (\text{Suc } k) cfs) - 1) = \text{length } cfs - 1$
 $(\text{is } \text{Suc } k + (\text{length } ?cfs - 1) = -)$
using B **and** C **by** *simp*
ultimately have $?P \models ?cfs \square$
using A **and** B **and** C **by** (*auto simp: execl-all-def intro: execl-drop*)
moreover assume $\forall cfs'. \text{length } cfs' < \text{length } cfs \longrightarrow ?P \models cfs' \square \longrightarrow ?Q \text{ cfs}'$
hence $\text{length } ?cfs < \text{length } cfs \longrightarrow ?P \models ?cfs \square \longrightarrow ?Q \text{ ?cfs } ..$
ultimately have $c\text{pred } (\text{WHILE } b \text{ DO } c) (cfs ! \text{Suc } k) (cfs ! (\text{length } cfs - 1))$
using B **and** C **and** E **by** *simp*
moreover assume $\text{bval } b \text{ s and } (c, s) \Rightarrow s'$
ultimately show $c\text{pred } (\text{WHILE } b \text{ DO } c) (0, s, stk) (cfs ! (\text{length } cfs - \text{Suc } 0))$
using D **by** (*auto simp: cpred-def*)
qed

lemma *ccomp-correct:*

$c\text{comp } c \models cfs \square \Longrightarrow c\text{pred } c (cfs ! 0) (cfs ! (\text{length } cfs - 1))$
proof (*induction c arbitrary: cfs, simp-all add: Let-def, frule-tac [4] bcomp-ccomp,*
frule-tac [3] ccomp-ccomp, auto)
fix $a \text{ x } cfs$
assume $A: \text{acom } a @ [\text{STORE } x] \models cfs \square$
hence $\exists k < \text{length } cfs. \text{apred } a (\text{off } [] (cfs ! 0)) (\text{off } [] (cfs ! k))$
by (*rule-tac execl-all-sub, auto simp: execl-all-def*)
moreover obtain $s \text{ stk}$ **where** $B: cfs ! 0 = (0, s, stk)$
using A **by** (*cases cfs ! 0, simp add: execl-all-def*)
ultimately obtain k **where** $C: k < \text{length } cfs$ **and**
 $D: cfs ! k = (\text{size } (\text{acom } a), s, \text{aval } a \text{ s } \# \text{stk})$
by (*auto simp: apred-def*)
hence $cfs ! \text{Suc } k = (\text{size } (\text{acom } a) + 1, s(x := \text{aval } a \text{ s}), \text{stk})$
using A **by** (*insert execl-next [of acom a @ [STORE x] cfs k],*
simp add: execl-all-def, drule-tac impI, auto)
thus $c\text{pred } (x ::= a) (cfs ! 0) (cfs ! (\text{length } cfs - \text{Suc } 0))$
using A **and** B **and** C **and** D **by** (*insert execl-last [of acom a @ [STORE x]*

```

    cfs Suc k], simp add: execl-all-def cpred-def, drule-tac impI, auto)
next
  fix c1 c2 cfs s s' t stk k
  assume ccomp c1 @ ccomp c2  $\models$  cfs $\square$  and  $k < \text{length } cfs$  and
    cfs ! k = (size (ccomp c1) + size (ccomp c2), s', stk)
  moreover assume (c1, s)  $\Rightarrow$  t and (c2, t)  $\Rightarrow$  s'
  ultimately show cpred (c1;; c2) (0, s, stk) (cfs ! (length cfs - Suc 0))
    by (insert execl-last, auto simp: execl-all-def cpred-def)
next
  fix b c1 c2 cfs s stk
  assume A: bcomp (b, False, size (ccomp c1) + 1) @ ccomp c1 @
    JMP (size (ccomp c2)) # ccomp c2  $\models$  cfs $\square$ 
    (is bcomp ?x @ ?cc1 @ - # ?cc2  $\models$  - $\square$ )
  let ?P = bcomp ?x @ ?cc1 @ [JMP (size ?cc2)]
  have  $\exists k < \text{length } cfs. \text{bpred } ?x (\text{off } [] (cfs ! 0)) (\text{off } [] (cfs ! k))$ 
    using A by (rule-tac execl-all-sub, auto simp: execl-all-def)
  moreover assume B:  $\neg \text{bval } b \ s$  and cfs ! 0 = (0, s, stk)
  ultimately obtain k where C:  $k < \text{length } cfs$  and D: cfs ! k = (size ?P, s, stk)
    by (force simp: bpred-def)
  assume  $\bigwedge cfs. ?cc2 \models cfs\square \implies \text{cpred } c2 (cfs ! 0) (cfs ! (\text{length } cfs - \text{Suc } 0))$ 
  hence  $\exists k' < \text{length } cfs. \text{cpred } c2 (\text{off } ?P (cfs ! k)) (\text{off } ?P (cfs ! k'))$ 
    using A and C and D by (rule-tac execl-all-sub [where P'' = [], auto])
  then obtain k' where k' < length cfs and case cfs ! k' of (pc', s', stk')  $\Rightarrow$ 
    pc' = size (?P @ ?cc2)  $\wedge$  (c2, s)  $\Rightarrow$  s'  $\wedge$  stk' = stk
    using D by (fastforce simp: cpred-def)
  thus cpred (IF b THEN c1 ELSE c2) (0, s, stk) (cfs ! (length cfs - Suc 0))
    using A and B by (insert execl-last, auto simp: execl-all-def cpred-def Let-def)
next
  fix b c1 c2 cfs s s' stk k
  assume A: bcomp (b, False, size (ccomp c1) + 1) @ ccomp c1 @
    JMP (size (ccomp c2)) # ccomp c2  $\models$  cfs $\square$ 
    (is ?cb @ ?cc1 @ ?i # ?cc2  $\models$  - $\square$ )
  assume B:  $k < \text{length } cfs$  and C: cfs ! k = (size ?cb + size ?cc1, s', stk)
  hence D: cfs ! Suc k = (size (?cb @ ?cc1 @ ?i # ?cc2), s', stk)
    (is - = (size ?P, -, -))
  using A by (insert execl-next [of ?P cfs k], simp add: execl-all-def,
    drule-tac impI, auto)
  assume bval b s and (c1, s)  $\Rightarrow$  s'
  thus cpred (IF b THEN c1 ELSE c2) (0, s, stk) (cfs ! (length cfs - Suc 0))
    using A and B and C and D by (insert execl-last [of ?P cfs Suc k],
    simp add: execl-all-def cpred-def Let-def, drule-tac impI, auto)
next
  fix c1 c2 cfs
  assume A: JMPND (size (ccomp c1) + 1) # ccomp c1 @
    JMP (size (ccomp c2)) # ccomp c2  $\models$  cfs $\square$ 
    (is JMPND (?n1 + 1) # ?cc1 @ JMP ?n2 # ?cc2  $\models$  - $\square$ )
  let ?P = JMPND (?n1 + 1) # ?cc1 @ [JMP ?n2]
  assume
    B:  $\bigwedge cfs. ?cc1 \models cfs\square \implies \text{cpred } c1 (cfs ! 0) (cfs ! (\text{length } cfs - \text{Suc } 0))$  and

```

$C: \bigwedge cfs. ?cc_2 \models cfs \square \implies \text{cpred } c_2 (cfs ! 0) (cfs ! (\text{length } cfs - \text{Suc } 0))$
from A **obtain** $s \text{ stk}$ **where** $D: cfs ! 0 = (0, s, \text{stk})$
by (*cases* $cfs ! 0$, *simp add: execl-all-def*)
with A **have** $cfs ! 1 = (1, s, \text{stk}) \vee cfs ! 1 = (?n_1 + 2, s, \text{stk})$
by (*insert execl-next* [*of ?P @ ?cc_2 cfs 0*], *simp add: execl-all-def*,
drule-tac impI, auto)
moreover {
assume $E: cfs ! 1 = (1, s, \text{stk})$
hence $\exists k < \text{length } cfs. \text{cpred } c_1 (\text{off } [hd ?P] (cfs ! 1)) (\text{off } [hd ?P] (cfs ! k))$
using A **and** B **by** (*rule-tac execl-all-sub, auto simp: execl-all-def*)
then obtain k **where** $k < \text{length } cfs$ **and** *case* $cfs ! k$ **of** $(pc', s', \text{stk}') \Rightarrow$
 $pc' = ?n_1 + 1 \wedge (c_1, s) \Rightarrow s' \wedge \text{stk}' = \text{stk}$
using E **by** (*fastforce simp: cpred-def*)
moreover from this have *case* $cfs ! \text{Suc } k$ **of** $(pc', s', \text{stk}') \Rightarrow$
 $pc' = ?n_1 + ?n_2 + 2 \wedge (c_1, s) \Rightarrow s' \wedge \text{stk}' = \text{stk}$
using A **by** (*insert execl-next* [*of ?P @ ?cc_2 cfs k*], *simp add: execl-all-def*,
drule-tac impI, auto)
ultimately have $\text{cpred } (c_1 \text{ OR } c_2) (cfs ! 0) (cfs ! (\text{length } cfs - \text{Suc } 0))$
using A **and** D **by** (*insert execl-last* [*of ?P @ ?cc_2 cfs Suc k*],
simp add: execl-all-def cpred-def split-def Let-def, drule-tac impI, auto)
}
moreover {
assume $E: cfs ! 1 = (?n_1 + 2, s, \text{stk})$
with A **and** C **have** $\exists k < \text{length } cfs. \text{cpred } c_2 (\text{off } ?P (cfs ! 1)) (\text{off } ?P (cfs ! k))$
by (*rule-tac execl-all-sub* [*where P'' = []*], *auto simp: execl-all-def*)
then obtain k **where** $k < \text{length } cfs$ **and** *case* $cfs ! k$ **of** $(pc', s', \text{stk}') \Rightarrow$
 $pc' = ?n_1 + ?n_2 + 2 \wedge (c_2, s) \Rightarrow s' \wedge \text{stk}' = \text{stk}$
using E **by** (*fastforce simp: cpred-def*)
with A **and** D **have** $\text{cpred } (c_1 \text{ OR } c_2) (cfs ! 0) (cfs ! (\text{length } cfs - \text{Suc } 0))$
by (*insert execl-last, auto simp: execl-all-def cpred-def Let-def*)
}
ultimately show $\text{cpred } (c_1 \text{ OR } c_2) (cfs ! 0) (cfs ! (\text{length } cfs - \text{Suc } 0)) \dots$
qed

Finally, the main compiler correctness theorem, expressed using predicate *exec*, is proven. First, $P \vdash cf \rightarrow^* cf'$ is shown to imply the existence of a nonempty list of configurations cfs such that $P \models cfs$, whose initial and final configurations match cf and cf' , respectively. Then, the main theorem is derived as a straightforward consequence of this lemma and of the previous lemma *ccomp-correct*.

lemma *exec-execl* [*dest!*]:

$P \vdash cf \rightarrow^* cf' \implies \exists cfs. P \models cfs \wedge cfs \neq [] \wedge hd \ cfs = cf \wedge last \ cfs = cf'$
by (*erule star.induct, force, erule exE, rule list.exhaust, blast,*
simp del: last.simps, rule exI, subst execl.simps(1), simp)

theorem *ccomp-exec*:

$c\text{comp } c \vdash (0, s, \text{stk}) \rightarrow^* (\text{size } (c\text{comp } c), s', \text{stk}') \implies (c, s) \Rightarrow s' \wedge \text{stk}' = \text{stk}$
by (*insert ccomp-correct, force simp: hd-conv-nth last-conv-nth execl-all-def cpred-def*)

end

References

- [1] G. Klein. Theory HOL-IMP.Compiler2 (included in the Isabelle2021 distribution). <https://isabelle.in.tum.de/website-Isabelle2021/dist/library/HOL/HOL-IMP/Compiler2.html>.
- [2] A. Krauss. *Defining Recursive Functions in Isabelle/HOL*. <https://isabelle.in.tum.de/website-Isabelle2021/dist/Isabelle2021/doc/functions.pdf>.
- [3] T. Nipkow. *A Tutorial Introduction to Structured Isar Proofs*. <https://isabelle.in.tum.de/website-Isabelle2011/dist/Isabelle2011/doc/isar-overview.pdf>.
- [4] T. Nipkow. *Programming and Proving in Isabelle/HOL*, Feb. 2021. <https://isabelle.in.tum.de/website-Isabelle2021/dist/Isabelle2021/doc/prog-prove.pdf>.
- [5] T. Nipkow and G. Klein. *Concrete Semantics with Isabelle/HOL*. Springer-Verlag, Mar. 2021. (Current version: <http://www.concrete-semantics.org/concrete-semantics.pdf>).
- [6] T. Nipkow, L. C. Paulson, and M. Wenzel. *Isabelle/HOL – A Proof Assistant for Higher-Order Logic*, Feb. 2021. <https://isabelle.in.tum.de/website-Isabelle2021/dist/Isabelle2021/doc/tutorial.pdf>.